Paper 188-2007

Using SAS® to Determine Sample Sizes for Traditional 2-Stage and Adaptive 2-Stage Phase II Cancer Clinical Trial Designs

Adrienne Groulx, Kyung-hee (Kelly) Moon, St. Clare Chung
Scian Services Inc, Etobicoke, Ontario, Canada

ABSTRACT

In a Phase II clinical trial, the primary objective is to determine the efficaciousness of a drug, such that decisions to proceed with further studies and development of the drug are warranted. To detect a measurable effect in an investigation product, a two-stage design is often used, whereby an interim look at the data may result in the decision to terminate the trial early, or proceed with the second stage of the study. The Simon's two-stage design to determine sample size is one that is widely used and accepted in Phase II cancer trials. A number of alternative methods that implement adaptive aspects to the design have also been published and used in trials. This paper will look specifically at Lin and Shih's adaptive method, which include extensions to Simon's optimal and minimax designs. Currently, PROC POWER in SAS v9 does not produce sample size calculations for two-stage designs, which led to the motivation behind this paper. The SAS codes for the traditional and adaptive designs have been separated into two macros, %SIMON and %ADAPTIVE, designed to produce multiple sample size outputs which satisfy user-specified constraints. This paper is intended for intermediate SAS users.

INTRODUCTION

If a biostatistician were to tell you what question is most frequently asked of them, it would likely be related to sample size and power calculations. Certainly for Phase II cancer trial designs, a number of online web tools and sample size software are available to help with the study design, where one could simply plug in numbers and have power calculations and sample sizes appear on screen. The biostatistician, however, should be able to validate these numbers by understanding how they are derived, and explain how and when they should be used. This is the reason for creating the macros %SIMON and %ADAPTIVE for two-stage Phase II designs. For the purposes of this paper, only single-arm designs are explored.

As Phase II oncology studies offer the first glance into the efficacy of an investigational drug, a common endpoint used in these trials is tumour response rate, as defined by the lead clinician. If the response rate exceeds the minimum criteria at the conclusion of the trial, then a Phase III trial may be considered.

The following notations will be used throughout the paper:

- p_0 : An unacceptable response rate.
- p_1 : The target response rate.
- α : The false-positive rate, α , is the probability of declaring the drug effective when the true response probability is p_0.
- β : The false-negative rate, β , is the probability of declaring the drug as not effective when its true response probability is the target response rate p_1.
- n: The total number of subjects to be evaluated.
- r: The upper-limit of the number of responses in n patients such that futility of the drug is concluded (ie. r+1 is the minimum number of responses in n patients that would warrant further development of the drug)

SINGLE STAGE DESIGN

In a single stage Phase II clinical trial, a pre-determined number of patients are treated with the investigational drug, and the response rate is measured. A critical value, typically found using binomial exact methods, is specified in

advance, such that if the number of responses is less than this critical value, than the drug is deemed ineffective in the population targeted.

Example 1:

Consider a SAS example of a single stage design where p_0=0.2, p_1 =0.4, and α=0.05. Using the ONESAMPLEFREQ statement in PROC POWER, we solve for N based on powers ranging from 75% to 99%:

```
proc power;
onesamplefreq TEST=ADJZ METHOD=NORMAL
  sides=1
   nullproportion = 0.20
   proportion = 0.40
   ntotal = .
   power = 0.75 to 0.99 by 0.01;
  plot x=power  xopts=(ref=.8 .85 .9 .95 crossref=yes);
run;
```

The graph to the right is produced, and can be used to determine the sample size which best corresponds to the desired power of the study. For example, in order to achieve 90% power to detect at least 20% improvement in response rate, we require that N_{total}=46.09 patients (or 47, as we would want to round up to an integer to achieve at least 90% power). Next, the critical value for the go/no-go decision rule is found by replacing the "TEST=ADJ METHOD=NORMAL" statement in the original SAS code to "METHOD=EXACT". By entering the value for N (ex. N_{total}=47) into our SAS code, we find that the critical upper value, C_U, is 15. That is, H_0: p≤ p_0 is rejected when the number of responses is greater than or equal to 15. Equivalently, we would conclude futility of the drug if fewer than 15 responses in 47 patients are observed.



Single stage phase II cancer clinical trials are generally not as favorable as two-stage designs. The reason for this comes from the ethical dilemmas that they pose. Namely, a single stage design runs the risk of exposing too many sick patients in need of efficacious drugs to potentially inactive and dangerous drugs. This risk was the motivation behind the creation of two-stage Phase II clinical trials.

SIMON'S OPTIMAL AND MINIMAX 2-STAGE DESIGN

In Simon's two-stage design, after a pre-determined number of patients have been treated, the trial is paused, and the response rate is evaluated. If a pre-specified minimal response rate has not been achieved in the first stage of the trial, it is concluded that the treatment is not worth pursuing and the trial is ended. Otherwise, enrollment continues until a pre-determined number of additional patients are accrued. At the conclusion of the clinical trial, the drug will be declared effective or ineffective.

Simon's optimal design is one which minimizes the expected sample size given a poor response rate. If the numbers of patients studied in the first and second stage are denoted by n_I and n_2 respectively, then the expected sample size is

$$E(N \mid p) = E(N) = n_I + (1 - PET)\, n_{2,}$$

where PET represents the probability of early termination after the first stage. If after the first stage, the number of responses is fewer than or equal to a pre-determined value, r_1 of n_1, then PET = $B(r_1; n_1, p)$, where B denotes the cumulative distribution function evaluated at r_1 for a random variable that is binomially distributed with parameters n_1 and p, the true probability of response. If the trial continues to Stage 2, then the drug is rejected at the end of this stage if r or fewer responses are observed. Hence the probability of rejecting a drug with success probability p is

$$B(r_1; n_1, p) + \sum_{x=r_1+1}^{\min[n_1, r]} b(x; n_1, p) B(r - x; n_2, p),$$

where b denotes the binomial probability mass function.

Simon's minimax design is the one that minimizes the maximum sample size, such that both the α and β constraints are satisfied.

The graph to the right is used to illustrate the differences between the two Simon designs. In this example, the following criteria were provided:

p_0=0.2
p_1=0.4
α =0.05
β =0.1



A further constraint of N≤ 80 was added in order to reduce processing times which are inherent as N increases. The points on the graph indicate the expected value of N under p_0, such that the α and ß constraints of the given sample size are satisfied, and E(N|p_0) is minimized (as multiple feasible solutions may exist for each value of N). The optimal design is easily detected, as it is the lowest point on the graph (ie. where E(N|p_0) is smallest). Thus, for Simon's optimal design, the overall sample size, n, is equal to 54. The minimum sample size which satisfies both the α and β constraints is at n=45, thus making it the appropriate sample size for Simon's minimax design.

To determine the sample sizes (ie. n_1, n) and required response rates (ie. r_1, r), the macro %SIMON was created using SAS. The macro is called as follows:

```
%SIMON(
    p0=          /*  Unacceptable response                                  */
    p1=          /*  Acceptable response                                    */
    alpha=       /*  Type I Error (probability of accepting a poor drug)    */
    beta=        /*  1-Power (probability of rejecting a good drug)         */
    usern=       /*  User-specified maximum sample size                     */
    onoff=       /*  'ON'=all feasible solutions, 'OFF'=only optimal solutions  */
    plot_all=    /*  Storage directory path for the Optimal and Minimax plot  */
    )
```

It is up to the trial clinicians to provide the acceptable and unacceptable response rates for their drug. The decision is typically based on historical controls, or appropriate expectations of the drug.

The first job of the macro is to determine whether the parameters entered in the macro call are valid. For example, the macro %SIMON will be stopped in each of the following situations, and a message will be sent to the log:

- The key parameters (p0, p1, alpha, beta, or usern) were not given, or are not in numeric format.
- The user-specified maximum sample size, usern, cannot exceed 120.
- The user-specified parameters will not yield a feasible solution.

If all of the above validation checks are satisfied, the macro will loop through all of the possible two-stage designs for testing Ho: $p \leq p_0$ vs. H_A: $p \geq p_1$, where N is subject to a user-specified cap, up to a maximum sample size of $N \leq 120$. The iterative steps used in the %SIMON macro code are summarized below:

1. Compute the first term in Simon's equation, ie. $B(r_1; n_1, p)$, for all solution sets (r1, n1, p). Eliminate those solutions which do not meet the pre-defined β constraint.

```
data stage1;
do n1=2 to &usern-1;
   do r1=0 to n1 while (r1<n1);
        term1_p0 = cdf('BINOMIAL', r1, &p0, n1);
        term1_p1 = cdf('BINOMIAL', r1, &p1, n1);
        if term1_p1=<&beta then output; /*remove solution sets that do
                                          not meet the beta requirement*/
     end;
end;
run;
```

2. Using the reduced solution sets found in the above procedure, compute the second term in the equation, ie. $\sum_{x=r_1+1}^{\min[n_1, r]} b(x; n_1, p) B(r-x; n_2, p)$, ensuring that the addition of this term in the formula continue to meet the α and β constraints.

```
data stage12;
set stage1;
 do n=n1+1 to &usern;
  do r=r1+1 to n while (r1<r<n);
   term2_p0=0; *initialize the summation terms for alpha & beta calculations;
   term2_p1=0;
     do x=r1+1 to min(r, n1);
      dum0=pdf('BINOMIAL', x, &p0, n1)*cdf('BINOMIAL', r-x, &p0, n-n1);
      dum1=pdf('BINOMIAL', x, &p1, n1)*cdf('BINOMIAL', r-x, &p1, n-n1);
      term2_p0= term2_p0+dum0;*recursive formulae used for summation terms;
      term2_p1= term2_p1+dum1;
     end;
   if 1-(term1_p0+term2_p0)=<&alpha and (term1_p1+term2_p1)=<&beta then
   output;
  end;
 end;
run;
```

The solution sets found in the data set, *stage12*, are sorted accordingly in order to determine the minimax and optimal solutions.

In the output produced by the %SIMON macro, the designs that satisfy Simon's optimal and minimax criteria are printed first, followed by all other feasible solutions which satisfy the bounds on Type I & II errors. Each design is output with its' early stopping criteria in which Ho is accepted (ie. where r_1 or fewer responses are seen in the first n_1 patients). Otherwise, the design specifies to continue to $n = n_1 + n_2$ patients, and reject Ho if more than r responses out of n patients are observed.

Example 2:

To compare Simon's two-stage design with the single stage design, the parameters used in Example 1 will be revisited: p_0=0.2, p_1=0.4, α =0.05, β =0.1. After calling the %SIMON macro with these conditions, the output below

is produced. The first few lines of the output contain the Single Stage, Minimax, and Optimal designs retrieved from the given parameters, and are followed by additional designs which satisfy the alpha and beta constraints. In this example, the list of feasible solutions is cut-off after the first 13 results.

Single Stage and Simon Two Stage Designs for p0=0.2, p1=0.4, Alpha=0.05, Beta=0.1

Constraints Satisfied	Alpha	Beta	r1	n1	n2	r	n	Expected Sample Size
Single Stage	0.037	0.099	--	--	--	14	47	47.00
Minimax	0.048	0.100	5	24	21	13	45	31.23
Optimal	0.048	0.096	4	19	35	15	54	30.43
All Feasible Solutions:								
*alpha & beta satisfied	0.049	0.100	3	18	27	13	45	31.47
*alpha & beta satisfied	0.048	0.099	6	27	18	13	45	32.16
*alpha & beta satisfied	0.050	0.094	4	22	23	13	45	32.51
*alpha & beta satisfied	0.050	0.094	5	25	20	13	45	32.67
*alpha & beta satisfied	0.049	0.098	2	15	30	13	45	33.06
*alpha & beta satisfied	0.050	0.094	3	19	26	13	45	33.17
*alpha & beta satisfied	0.050	0.093	6	28	17	13	45	33.47
*alpha & beta satisfied	0.048	0.097	7	30	15	13	45	33.59
*alpha & beta satisfied	0.050	0.091	7	31	14	13	45	34.78
*alpha & beta satisfied	0.049	0.094	8	33	12	13	45	35.40
*alpha & beta satisfied	0.047	0.098	9	35	10	13	45	36.46
*alpha & beta satisfied	0.049	0.092	9	36	9	13	45	37.51
*alpha & beta satisfied	0.048	0.094	10	38	7	13	45	38.85
... etc ...								

Based on the results above, the two-stage minimax and optimal designs will first assess the efficacy/futility of a drug after n_1=24 and n_1=19 subjects, respectively, are enrolled and evaluated. A single-stage design, however, will enroll up to 47 patients before determining futility.

LIN AND SHIH'S ADAPTIVE 2-STAGE DESIGN

The adaptive two stage design offers added benefits to the traditional two-stage design. By using the information at the end of the first stage of the study, the original assumption of the response rate can be reassessed in the event that is was too optimistic or too skeptical to the true response rate. The second stage of the trial can be adjusted accordingly, while still controlling the Type I error rate.

What are the advantages to using a two-stage adaptive design?
- Possible reduction in sample size compared to traditional two-stage.
- Expedite the decision process.
- Flexibility within the trial design.

A disadvantage to the two-stage adaptive design arises in the total sample size calculation. The Stage 2 portion of the trial is dependent on the results observed at the completion of Stage 1, which means that the total number of patients to be enrolled is not firmly set at the protocol development stage. This could pose a problem when budgeting for the trial. For costing purposes, Lin and Shih's adaptive design can estimate the number of patients needed for the target response rates assumed. When designing a protocol which incorporates Lin and Shih's adaptive design, one needs to pre-specify all possible actions to be taken once Stage 1 is completed.

To implement Lin and Shih's method, additional notation needs to be introduced. The high and low choices of the target response rate are denoted by p_1 and p_2, respectively, where $p_0<p_1 \leq p_2$. If x represents the number of observed responses in the first stage out of n_1 patients, then the procedure is as follows:

- If $x \leq s_1$, conclude futility of the drug, and the trial is stopped.
- If $s_1< x \leq r_1$, then the study is powered at $(1-ß_1)$ for H_A: $p \geq p_1$, and a total of m (where $m=n_1+m_2$) subjects are evaluated. The treatment is rejected if s or fewer patients in m respond.
- If $x > r_1$, then the study is powered at $(1-ß_2)$ for H_A: $p \geq p_2$, and a total of n (where $n=n_1+n_2$) subjects are evaluated. The treatment is rejected if r or fewer patients in n respond.

Based on this design, the probability of rejecting a drug with success probability p is

$$B(s_1;n_1,p)+\sum_{x=s_1+1}^{\min[r_1,s]} b(x;n_1,p)B(s-x;m_2,p)+\sum_{y=r_1+1}^{\min[n_1,r]} b(y;n_1,p)B(r-y;n_2,p) \qquad [1]$$

A solution set (s1, r1, n1, s, m, r, n, p) is considered to be feasible if [1] satisfies the following constraints:

- [1] ≥ 1 – α, for p≤ p_0,
- [1] ≤ β_1, for p≥ p_1,
- [1] ≤ β_2, for p≥ p_2.

To find such feasible solutions, the following call the %ADAPTIVE code is implemented:

```
%ADAPTIVE(
    p0=           /*  Unacceptable response                                  */
    p1=           /*  Acceptable response rate                               */
    p2=           /*  Alternative acceptable response rate                   */
    alpha=        /*  Type I Error                                           */
    beta1=        /*  Probability of rejecting a good drug under HA: p≥ p1   */
    beta2=        /*  Probability of rejecting a good drug under HA: p≥ p2   */
    usern=        /*  User-specified maximum sample size                     */
    all=          /*  Output dataset name                                    */
    )
```

A key difference in the call of this macro compared to the %SIMON macro is that two new parameters, p_2 and ß$_2$, have been added. Once the program is run, the following 4 optimal designs discussed in Lin and Shih's paper are produced:

O1: E(N_0) is smallest
O2: max{E(N_i) | *i=0,1,2*} is smallest
O3: max (n,m) is smallest among feasible solutions, and E(N_0) is smallest among such solutions
O4: max (n,m) is smallest among feasible solutions, and max{E(N_i) | i=0,1,2} is smallest among such solutions,

where E(N_i)=E(N | p_i) for i=0,1,2, and where m and n are the total sample sizes under p_1 and p_2, respectively.

The following excerpt of code from the %ADAPTIVE macro is used to determine Optimal Types 1-4 after the set of all feasible solutions is established. The method of solving for the solution sets is similar to the iterative steps shown in the sample code for the %SIMON macro. In the code below, the dataset, *type*, contains all feasible solutions from the user-specified parameters.

```
/*Optimality Type 1 (O1):  E(N0) is smallest*/
proc sql noprint;
create table type1 as
select alpha, beta1, beta2, s1, r1, n1, s, m, r, n, en0, en1, en2,
          minmax_en, minmax_nm, max_en, max_nm
from type where min_en0=en0;
quit;

/*Optimality Type 2 (O2): Max{E(N0), E(N1), E(N2)} is smallest*/
proc sql noprint;
create table type2 as
select alpha, beta1, beta2, s1, r1, n1, s, m, r, n, en0, en1, en2,
          minmax_en, minmax_nm, max_en, max_nm
from type where max_en=minmax_en;
quit;

/*For Optimality Types 3 and 4, first determine when max(n, m) is smallest*/
proc sql noprint;
create table type_ as
select alpha, beta1, beta2, s1, r1, n1, s, m, r, n, en0, en1, en2,
```

```
          max_en, min(en0) as nm_min_en0, min(max_en) as nm_minmax_en
from type where max_nm=minmax_nm;
quit;

/*Optimality Type 3 (O3): max(n, m) is smallest, and E(N0) is smallest*/
proc sql noprint;
create table type3 as
select alpha, beta1, beta2, s1, r1, n1, s, m, r, n, en0, en1, en2
from type_ where nm_min_en0=en0;
quit;

/*Optimality Type 4 (O4): max(n, m) is smallest, and Max{E(N0), E(N1), E(N2)} is
smallest*/
proc sql noprint;
create table type4 as
select alpha, beta1, beta2, s1, r1, n1, s, m, r, n, en0, en1, en2,
          nm_min_en0, nm_minmax_en
from type_ where max_en=nm_minmax_en;
quit;
```

Example 3:

Suppose the model parameters used in Examples 1 and 2 are similarly applied to the adaptive design. Due to the additional parameters required in Lin and Shih's method, the assumptions will vary slightly.
That is: p_0=0.2, p_1=0.35, p_2=0.4, α =0.05, β_1 =0.2, and β_2 =0.1.

After calling the %ADAPTIVE macro with these conditions, the output below is produced. It is noted that Optimal Type 3 and 4 are the same.

Lin and Shih Two Stage Design for p0=0.20, p1=0.35, p2=0.40, Alpha=0.05, Beta1=0.2, Beta2=0.1

Optimal	Design Parameters							True			Expected Sizes		
Type	s1	r1	n1	s	m	r	n	Alpha	Beta1	Beta2	EN0	EN1	EN2
O1	5	6	23	12	45	20	74	0.050	0.200	0.070	34.352	63.770	69.218
O2	4	8	24	18	63	9	28	0.050	0.198	0.064	43.799	44.754	38.953
O3	6	12	31	15	53	13	40	0.050	0.200	0.058	40.379	48.556	46.477
O4	6	12	31	15	53	13	40	0.050	0.200	0.058	40.379	48.556	46.477

Suppose we wish to compare Lin and Shih's designs with Simon's optimal and minimax designs. In the comparisons below, we will focus only on Lin and Shih's Optimal Types 1 and 3, as they are extension of Simon's optimal and minimax designs, respectively.

Comparison 1: Simon's Optimal Design with Lin and Shih's Optimal Type 1

In Simon's optimal design, the first stage requires that 19 subjects be evaluated, and a minimum of 5 responses observed in order to proceed to the second stage. Lin and Shih's O1 design require a minimum of 6 responses in 23 patients in order to proceed to the second stage. When 6 responses are observed, Lin and Shih's design suggests that the more conservative target response of 35% be chosen over the 40% target response. By lowering the response rate, fewer subjects are needed (N=45, versus N=54 in Simon's design), however, there are two prices to pay for this decrease in sample size: (1) the study is powered at 80%, instead of 90%, and (2) the lower target response rate of 35% is not as clinically desirable as a response rate of 40%.

In the event that greater than 6 responses are observed in Stage 1 of Lin and Shih's design, the target response rate of 40% is chosen, and a total of 74 subjects are enrolled (versus the N=54 required in Simon's design). Although the power is more attractive in Lin and Shih's design (93% power versus 90% power in Simon's design), the difference in the total sample size required is the price to pay for the ability to "hedge your bets" after Stage 1.

Comparison 2: Simon's Minimax Design with Lin and Shih's Optimal Type 3

The number of subjects needed in the first stage of Simon's minimax design and Lin and Shih's O3 design are n_1=24 and n_1=31, respectively, with minimum number of responses of 6 and 7, respectively. In the event that 7 to 12 responses are observed in Lin and Shih's design, the response rate of 35% is chosen, and a total of 53 subjects are enrolled (versus 45 in Simon's minimax design). That is, in this example, Lin and Shih require that a greater number of subjects be enrolled in order to proceed to the second stage, but at a reduced power of 80%, and a reduced target response rate of 35%. This is the cost of having the ability to hedge your bet in the minimax scenario.

On the other hand, if more than 12 subjects respond in the first stage of Lin and Shih's O3 design, the more desirable response rate of 40% is chosen, and a total of N=40 subjects (vs. N=45 in Simon's minimax design) would be enrolled. The power of this study is 94% (compared to 90% in Simon's design).

DISCUSSION

Due to the extensive iterative nature of the algorithms needed to solve for feasible solution sets, the %SIMON and %ADAPTIVE macros were both limited to computing up to a maximum sample size of N=120 for the purposes of this paper. An additional constraint is also imposed on the minimum sample size needed in the first stage (ie. $n_1 \geq 2$). Even with the capped sample size imposed on the %ADAPTIVE macro, run-times of up to 5 hours can be expected, depending on the user-specified N.

Practical limitations are also imposed on the α and β parameters, in order to maintain the integrity of the trial. That is, the Type I and Type II error rates are restricted to $0 < \alpha \leq 0.1$, and $0 < \beta \leq 0.2$, respectively. In the %ADAPTIVE macro, a further constraint of $\beta_1 \leq \beta_2$ is imposed.

A comment about the output produced by the %ADAPTIVE macro in Example 3: Although identical parameters from Lin and Shih's paper were used, the %ADAPTIVE output shown in this paper produced different solutions for Optimality Types 1 and 2 than those printed in Lin and Shih's paper. The reason for this difference stems from the limitations imposed on the total sample size calculation used by Lin and Shih. That is, the maximum sample size must be within 0.85 – 1.5 times that of a single-stage design. In Example 3, the single-stage design calls for a sample size of 47 patients, thereby imposing an upper limit under p_2 of n=71 patients. Since the %ADAPTIVE macro does not impose these limits, the higher sample size of n=74 was found to have a lower EN_0, as required by the O1 condition (ie. EN_0=34.35 in the %ADAPTIVE output vs EN_0=34.67 in Lin and Shih's output).

CONCLUSION

The ability to use pre-coded functions in SAS is a convenience that we sometimes take for granted. Future versions of SAS might include sample sizes solutions for various two-stage designs within the PROC POWER function, but in the meantime, a trial statistician must know how to solve for, and justify the sample size calculations used in a Phase II cancer trial design. By involving a statistician in the protocol design discussions, it is no longer necessary to use the stand-by Phase II designs of past.

With new and innovative adaptive methods being introduced into multi-stage designs to expedite the go-no-go decision processes, an increasing trend to embrace these methods can be observed throughout the industry. Incorporating adaptive methods into the trial design can be a great solution for those who would like the option of choosing the best direction to take once some data is known. However, as was observed in Example 3, there are potential penalties to pay for this safety net. It is the responsibility of the trial statistician to inform clinicians of the most appropriate designs for their upcoming two-stage trial, and ensure that the risks and benefits of one design over another is understood before finalizing a protocol.

REFERENCES

(1) Simon R (1989), *Optimal Two-Stage Designs for Phase II Clinical Trials*, Controlled Clinical Trials 10:1-10.

(2) Y. Lin and W.J. Shih (2004), *Adaptive Two-Stage Designs for Single-Arm Phase IIA Cancer Clinical Trials*, Biometrics 60: 482-490 (June 2004)

(3) Jung SH, Lee T, Kim KM, George SL. (2004) *Admissible two-stage designs for phase II cancer clinical trials*. Statistics in Medicine 2004; 23: 561 - 569

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author(s) at:

Adrienne Groulx
Scian Services Inc
Etobicoke, Ontario Canada
E-mail: agroulx@scian.com
Web: www.scian.com

Kyung-hee (Kelly) Moon
Scian Services Inc
Etobicoke, Ontario Canada
E-mail: kmoon@scian.com
Web: www.scian.com

St. Clare Chung
Scian Services Inc
Etobicoke, Ontario Canada
E-mail: schung@scian.com
Web: www.scian.com